

NAG Library Routine Document

D06AAF

Note: before using this routine, please read the Users' Note for your implementation to check the interpretation of ***bold italicised*** terms and other implementation-dependent details.

1 Purpose

D06AAF generates a triangular mesh of a closed polygonal region in $\mathbb{R}^2$, given a mesh of its boundary. It uses a simple incremental method.

2 Specification

```

SUBROUTINE D06AAF(NVB, NVMAX, NEDGE, EDGE, NV, NELT, COOR, CONN, BSPACE,
1          SMOOTH, COEF, POWER, ITRACE, RWORK, LRWORK, IWORK,
2          LIWORK, IFAIL)

  INTEGER          NVB, NVMAX, NEDGE, EDGE(3,NEDGE), NV, NELT,
1          CONN(3,2*(NVMAX-1)), ITRACE, LRWORK, IWORK(LIWORK),
2          LIWORK, IFAIL
  double precision COOR(2,NVMAX), BSPACE(NVB), COEF, POWER, RWORK(LRWORK)
  LOGICAL          SMOOTH

```

3 Description

D06AAF generates the set of interior vertices using a process based on a simple incremental method. A smoothing of the mesh is optionally available. For more details about the triangulation method, consult the D06 Chapter Introduction as well as George and Borouchaki (1998).

This routine is derived from material in the MODULEF package from INRIA (Institut National de Recherche en Informatique et Automatique).

4 References

George P L and Borouchaki H (1998) *Delaunay Triangulation and Meshing: Application to Finite Elements* Editions HERMES, Paris

5 Parameters

- | | | |
|----|--|--------------|
| 1: | NVB – INTEGER | <i>Input</i> |
| | <i>On entry:</i> the number of vertices in the input boundary mesh. | |
| | <i>Constraint:</i> $3 \leq \text{NVB} \leq \text{NVMAX}$. | |
| 2: | NVMAX – INTEGER | <i>Input</i> |
| | <i>On entry:</i> the maximum number of vertices in the mesh to be generated. | |
| 3: | NEDGE – INTEGER | <i>Input</i> |
| | <i>On entry:</i> the number of boundary edges in the input mesh. | |
| | <i>Constraint:</i> $\text{NEDGE} \geq 1$. | |

- 4: EDGE(3,NEDGE) – INTEGER array Input
On entry: the specification of the boundary edges. EDGE(1,*j*) and EDGE(2,*j*) contain the vertex numbers of the two end points of the *j*th boundary edge. EDGE(3,*j*) is a user-supplied tag for the *j*th boundary edge and is not used by D06AAF.
Constraint: $1 \leq \text{EDGE}(i,j) \leq \text{NVB}$ and $\text{EDGE}(1,j) \neq \text{EDGE}(2,j)$, for $i = 1, 2$ and $j = 1, 2, \dots, \text{NEDGE}$.
- 5: NV – INTEGER Output
On exit: the total number of vertices in the output mesh (including both boundary and interior vertices). If NVB = NVMAX, no interior vertices will be generated and NV = NVB.
- 6: NELT – INTEGER Output
On exit: the number of triangular elements in the mesh.
- 7: COOR(2,NVMAX) – **double precision** array Input/Output
On entry: COOR(1,*i*) contains the *x* co-ordinate of the *i*th input boundary mesh vertex; while COOR(2,*i*) contains the corresponding *y* co-ordinate, for $i = 1, \dots, \text{NVB}$.
On exit: COOR(1,*i*) will contain the *x* co-ordinate of the (*i* – NVB)th generated interior mesh vertex; while COOR(2,*i*) will contain the corresponding *y* co-ordinate, for $i = \text{NVB} + 1, \dots, \text{NV}$. The remaining elements are unchanged.
- 8: CONN(3,2 × (NVMAX – 1)) – INTEGER array Output
On exit: the connectivity of the mesh between triangles and vertices. For each triangle *j*, CONN(*i*,*j*) gives the indices of its three vertices (in anticlockwise order), for $i = 1, 2$ and 3, and $j = 1, \dots, \text{NELT}$.
- 9: BSPACE(NVB) – **double precision** array Input
On entry: the desired mesh spacing (triangle diameter, which is the length of the longer edge of the triangle) near the boundary vertices.
Constraint: BSPACE(*i*) > 0.0, for $i = 1, 2, \dots, \text{NVB}$.
- 10: SMOOTH – LOGICAL Input
On entry: indicates whether or not mesh smoothing should be performed.
 If SMOOTH = .TRUE., the smoothing is performed; otherwise no smoothing is performed.
- 11: COEF – **double precision** Input
On entry: the coefficient in the stopping criteria for the generation of interior vertices. This parameter controls the triangle density and the number of triangles generated is in $O(\text{COEF}^2)$. The mesh will be finer if COEF is greater than 0.7165 and 0.75 is a good value.
Suggested value: 0.75.
- 12: POWER – **double precision** Input
On entry: controls the rate of change of the mesh size during the generation of interior vertices. The smaller the value of POWER, the faster the decrease in element size away from the boundary.
Suggested value: 0.25.
Constraint: $0.1 \leq \text{POWER} \leq 10.0$.

- 13: ITRACE – INTEGER *Input*
On entry: the level of trace information required from D06AAF.
 ITRACE ≤ 0
 No output is generated.
 ITRACE ≥ 1
 Output from the meshing solver is printed on the current advisory message unit (see X04ABF). This output contains details of the vertices and triangles generated by the process.
 You are advised to set ITRACE = 0, unless you are experienced with finite element mesh generation.
- 14: RWORK(LRWORK) – *double precision* array *Workspace*
 15: LRWORK – INTEGER *Input*
On entry: the dimension of the array RWORK as declared in the (sub)program from which D06AAF is called.
Constraint: LRWORK $\geq$ NVMAX.
- 16: IWORK(LIWORK) – INTEGER array *Workspace*
 17: LIWORK – INTEGER *Input*
On entry: the dimension of the array IWORK as declared in the (sub)program from which D06AAF is called.
Constraint: LIWORK $\geq 16 \times \text{NVMAX} + 2 \times \text{NEDGE} + \max(4 \times \text{NVMAX} + 2, \text{NEDGE}) - 14$.
- 18: IFAIL – INTEGER *Input/Output*
On entry: IFAIL must be set to 0, -1 or 1. If you are unfamiliar with this parameter you should refer to Section 3.3 in the Essential Introduction for details.
On exit: IFAIL = 0 unless the routine detects an error (see Section 6).
 For environments where it might be inappropriate to halt program execution when an error is detected, the value -1 or 1 is recommended. If the output of error messages is undesirable, then the value 1 is recommended. Otherwise, if you are not familiar with this parameter, the recommended value is 0. **When the value -1 or 1 is used it is essential to test the value of IFAIL on exit.**

6 Error Indicators and Warnings

If on entry IFAIL = 0 or -1, explanatory error messages are output on the current error message unit (as defined by X04AAF).

Errors or warnings detected by the routine:

IFAIL = 1

On entry, NVB < 3 or NVB > NVMAX,
 or NEDGE < 1,
 or $\text{EDGE}(i, j) < 1$ or $\text{EDGE}(i, j) > \text{NVB}$, for some $i = 1, 2$ and $j = 1, 2, \dots, \text{NEDGE}$,
 or $\text{EDGE}(1, j) = \text{EDGE}(2, j)$, for some $j = 1, 2, \dots, \text{NEDGE}$,
 or $\text{BSPACE}(i) \leq 0.0$, for some $i = 1, 2, \dots, \text{NVB}$,
 or $\text{POWER} < 0.1$ or $\text{POWER} > 10.0$,
 or $\text{LIWORK} < 16 \times \text{NVMAX} + 2 \times \text{NEDGE} + \max(4 \times \text{NVMAX} + 2, \text{NEDGE}) - 14$,
 or LRWORK < NVMAX.

IFAIL = 2

An error has occurred during the generation of the interior mesh. Check the definition of the boundary (arguments COOR and EDGE) as well as the orientation of the boundary (especially in

the case of a multiple connected component boundary). Setting ITRACE > 0 may provide more details.

7 Accuracy

Not applicable.

8 Further Comments

The position of the internal vertices is a function of the positions of the vertices on the given boundary. A fine mesh on the boundary results in a fine mesh in the interior. The algorithm allows you to obtain a denser interior mesh by varying NVMAX, BSPACE, COEF and POWER. But you are advised to manipulate the last two parameters with care.

You are advised to take care to set the boundary inputs properly, especially for a boundary with multiply connected components. The orientation of the interior boundaries should be in **clockwise** order and opposite to that of the exterior boundary. If the boundary has only one connected component, its orientation should be **anticlockwise**.

9 Example

In this example, a geometry with two holes (two interior circles inside an exterior one) is meshed using the simple incremental method (see the D06 Chapter Introduction). The exterior circle is centred at the origin with a radius 1.0, the first interior circle is centred at the point $(-0.5, 0.0)$ with a radius 0.49, and the second one is centred at the point $(-0.5, 0.65)$ with a radius 0.15. Note that the points $(-1.0, 0.0)$ and $(-0.5, 0.5)$ are points of 'near tangency' between the exterior circle and the first and second circles.

The boundary mesh has 100 vertices and 100 edges (see Figure 1). Note that the particular mesh generated could be sensitive to the *machine precision* and therefore may differ from one implementation to another. Figure 2 contains the output mesh.

9.1 Program Text

```
*      D06AAF Example Program Text
*      Mark 20 Release. NAG Copyright 2001.
*      .. Parameters ..
      INTEGER          NIN, NOUT
      PARAMETER        (NIN=5,NOUT=6)
      INTEGER          NBEDMX, NVMAX, NELTMAX, LIWORK, LRWORK
      PARAMETER        (NBEDMX=100,NVMAX=250,NELTMAX=2*(NVMAX-1),
+                      LIWORK=16*NVMAX+2*NBEDMX+(4*NVMAX+2)-14,
+                      LRWORK=NVMAX)
*      .. Local Scalars ..
      DOUBLE PRECISION COEF, POWER
      INTEGER          I, I1, IFAIL, ITRACE, K, NEDGE, NELT, NV, NVB,
+                      REFTK
      LOGICAL          SMOOTH
      CHARACTER        PMESH
*      .. Local Arrays ..
      DOUBLE PRECISION BSPACE(NVMAX), COOR(2,NVMAX), RWORK(LRWORK)
      INTEGER          CONN(3,NELTMAX), EDGE(3,NBEDMX), IWORK(LIWORK)
*      .. External Subroutines ..
      EXTERNAL        D06AAF
*      .. Executable Statements ..
      WRITE (NOUT,*) 'D06AAF Example Program Results'
      WRITE (NOUT,*)

*
*      Skip heading in data file
*
      READ (NIN,*)

*
*      Reading of the geometry
*      Coordinates of the boundary mesh vertices and
*      edges references.
```

```

*      READ (NIN,*) NVB, NEDGE
*
*      IF (NVB.GT.NVMAX .OR. NEDGE.GT.NBEDMX) THEN
*        WRITE (NOUT,*) 'Problem with the array dimensions '
*        WRITE (NOUT,99999) ' NVB MAX ', NVB, NVMAX
*        WRITE (NOUT,99999) ' NEDGE MAX ', NEDGE, NBEDMX
*        GO TO 120
*      END IF
*
*      DO 20 I = 1, NVB
*        READ (NIN,*) I1, COOR(1,I1), COOR(2,I1)
20    CONTINUE
*
*      Boundary edges
*
*      DO 40 I = 1, NEDGE
*        READ (NIN,*) I1, EDGE(1,I1), EDGE(2,I1), EDGE(3,I1)
40    CONTINUE
*
*      Initialise mesh control parameters
*
*      DO 60 I = 1, NVB
*        BSPACE(I) = 0.05D0
60    CONTINUE
*      SMOOTH = .TRUE.
*      ITRACE = 0
*      COEF = 0.75D0
*      POWER = 0.25D0
*
*      Call to the mesh generator
*
*      IFAIL = 1
*
*      CALL D06AAF(NVB,NVMAX,NEDGE,EDGE,NV,NELT,COOR,CONN,BSPACE,SMOOTH,
+      COEF,POWER,ITRACE,RWORK,LRWORK,IWORK,LIWORK,IFAIL)
*
*      READ (NIN,*) PMESH
*
*      IF (IFAIL.NE.0) THEN
*        WRITE (NOUT,*)
*        WRITE (NOUT,99994) ' ** D06AAF returned with IFAIL = ', IFAIL
*        GO TO 120
*      ELSE IF (PMESH.EQ.'N') THEN
*        WRITE (NOUT,99998) 'NV   =', NV
*        WRITE (NOUT,99998) 'NELT =', NELT
*      ELSE IF (PMESH.EQ.'Y') THEN
*
*        Output the mesh to view it using the NAG Graphics Library
*
*        WRITE (NOUT,99997) NV, NELT
*        DO 80 I = 1, NV
*          WRITE (NOUT,99996) COOR(1,I), COOR(2,I)
80      CONTINUE
*
*        REFTK = 0
*        DO 100 K = 1, NELT
*          WRITE (NOUT,99995) CONN(1,K), CONN(2,K), CONN(3,K), REFTK
100     CONTINUE
*        ELSE
*          WRITE (NOUT,*) 'Problem with the printing option Y or N'
*        END IF
*
*      120 CONTINUE
*
*      99999 FORMAT (1X,A,2I6)
*      99998 FORMAT (1X,A,I6)

```

```

99997 FORMAT (1X,2I10)
99996 FORMAT (2(2X,E12.6))
99995 FORMAT (1X,4I10)
99994 FORMAT (1X,A,I5)
      END

```

9.2 Program Data

Note 1: since the data file for this example is quite large only a section of it is reproduced in this document. The full data file is distributed with your implementation.

```

D06AAF Example Program Data
      100      100      :NVB NEDGE
      1  0.100000E+01  0.000000E+00
      .
      .
      .
      100  -.353278E+00  0.681187E+00  :(I1, COOR(:,I),I=1,...,NVB)
      1  1  2  1
      .
      .
      .
      99  99 100  1
      100 100 71  1  :(I1, EDGE(:,I), I=1,...,NEDGE)
      'N'           :Printing option 'Y' or 'N'

```

9.3 Program Results

D06AAF Example Program Results

```

NV   =   250
NELT =   402

```




Figure 1
The boundary mesh of the geometry with two holes




Figure 2
Interior mesh of the geometry with two holes
